
�Modelle

ftRobs/ftRobsTS : Großmodell mit allen Robots





� EINFÜGENGRAFIK C:\\UM\\FISCHER\\FTROBSTS\\BILDER\\FTPLANT.BMP \* FORMATVERBINDEN \d ���

Modell ftRobsTS : Simultanes Zusammenspiel aller Robots des Konstruktionskastens "Industry Robots". Visual Basic 5 Programm unter Nutzung der FishF532.DLL.



Das Großmodell besteht aus den vier Robots des Kastens, die über Alu-Profile miteinander verbunden sind. Auf dem Interfacekasten des Schwenkroboters wurde ein Tonnen-Magazin installiert.

�

� EINFÜGENGRAFIK C:\\UM\\FISCHER\\FTROBSTS\\BILDER\\FTPLANTW.GIF \* FORMATVERBINDEN \d ���

ftRobsTS : Bedien und Anzeige-Panel

Hauptfunktionen

START / ENDE : Starten und Beenden des Programms. Beim Start wird die Verbindung zu den Interfaces hergestellt. Die Zuordnung Rob - Interface erfolgt über das INI-File ftRobs.INI.

HOME : gleichzeitiges Anfahren der Home-Positionen durch alle Robots.

Abbrechen / Programm / STOP : Abbrechen des Programms vor Ausführung der START-Funktion, Starten des im Script-File (Default : ftRobs.Rbs) festgelegten Programmablaufs. Der Programmablauf kann über STOP vorzeitig abgebrochen werden.

Wiederholen / Beenden : Anwahl bzw. Beenden des Endlosbetriebes. Muß vor Start des Programmablaufs über den Programm-Button erfolgen.

Die Buttons wechseln abhängig von der Programmsituation ihre Beschriftung und Funktion. Aus dem gleichen Grund sind sie zeitweise gesperrt.

Über Datei | Öffnen können alternative Scriptfiles geladen werden.

�Der Programmablauf besteht aus folgenden Schritten :

Säulen-Roboter : Holen der ersten Tonne aus dem Magazin, abstellen auf dem rechten Teller des Schwenk-Roboters. 

Schweiß-Roboter : Schweißen rechts. 

Schwenk-Roboter : Schwenken nach links. 

Schweiß-Roboter : Schweißen links. 

Knickarm-Roboter : Greifen der Tonne auf Teller links und Abstellen im Magazin-Eingangsbereich. 

Knickarm-Roboter : Schieben aller Tonnen des Magazins bis die rechte Tonne auf der Hol-Position steht. 

Synchronisation :

Erstellen von vier Threads für die einzelen Robots (CreateThread) und Anlegen der zugehörenden Ereignis-Objekte für Synchronisation und Fertigmeldung (CreateEvent, nicht signalisiert). 

Signal Event 3 (Programm-Button) : Start Säulen-Roboter 

Signal Event 2 (Säulen-Roboter) : Start Schweiß-Roboter 

Signal Event 1 (Schweiß-Roboter) : Start Schwenk-Roboter 

Signal Event 2 (Schwenk-Roboter) : erneuter Start Schweiß-Roboter 

Signal Event 4 (Schweiß-Roboter) : Start Knickarm-Roboter 

Signal Event 3 (Knickarm-Roboter) : Start Säulen-Roboter�(nur bei Endlosbetrieb) 

Die Robots laufen zeitweise simultan und beenden ihren Thread selber (Event Fertigmeldung), wenn ihre Aufgabe abgeschlossen ist. Im Endlosbetrieb wird der Thread nicht beendet, es wird auf ein neues Synchronisations Event gewartet (WaitForSingleObject, Event Synchronisation).

Der Primär-Thread (Programm-Button) wartet auf das Ende aller Robot-Threads (WaitForMultipleObjects, Events Fertigmeldung). Der jeweilige Robot-Status wird aus den Robot-Threads selber auf der Main-Form angezeigt. Dazu muß der Primär-Thread unterbrechbar sein (Warten auf Ereignis und Zeit in einer Schleife).

�Scriptfiles

Die Steuerung des Programmablaufs erfolgt über Scriptfiles, die mit NotePad erstellt werden können. Die Eingabedaten können mit einem TeachIn-Programm (z.B. ftTeach) ermittelt werden. Hinzu kommen Synchronisationspunkte zu den anderen Robots.

Aufbau  

2  1   152    60     56     0     zur Tonne links II

2  3   2      0      0      0     Rob2 parken     ---

2  3   4      1      0      0     Rob4 starten    ---

Robotnr für den die Zeile gilt

Auftragsart : 1 = Fahrbefehl, 3 = Synchronisationsbefehl

Wert1 - Wert4 : Parameter für die Befehle�Fahrbefehl Position von Säule, Arm waagerecht, Arm senkrecht, Greifer / Elektrode�Synchronistation : Wert1 = RobNr , Wert2 = 1 Rob starten, = 0 Rob parken.

Kommentar, der in der Statuszeile des Anzeige Panels erscheint. Muß eingegeben werden, es reicht ein - Bindestrich.

Im Anzeige Panel wird außerdem die aktuelle Zeilennummer des Scriptfiles angezeigt. Die Befehle für die Robots werden in einem gemeinsamen Scriptfile in Reihenfolge der Robot-Nummern abgelegt. Es können Robots fehlen. Rob1 kann nicht über ein Scriptfile gesteuert werden, er hat ein festes Programm.

Anforderungen :

Die vier Robot-Modelle, wie im Original-Handbuch beschrieben. 

Anschluß über serielle Interfaces (prinzipiell auch über parallele Interfaces möglich (nicht getestet). 

Multiport-Karte (verwendet wurde VS Vision System PCI-400) mit den Anschlüssen COM5 - COM8 (oder anderen). 

Visual Basic 5.0 (SP3). 

Und gute Nerven beim Test : Es kann nicht in der Entwicklungsumgebung getestet werden sondern nur mit EXE-Files. Standard Fehlermeldung : Katastrophe, Abbruch, Hersteller fragen. Anschließend den Wust der verhakten Robots entwirren und restaurieren. Aber wenns erst mal rundläuft gehts ganz gut.



�Anmerkungen zur MultiThread-Programmierung bei ftRobsTS





Deklarationen

Public Declare Function CloseHandle Lib "kernel32" _

  (ByVal hObject As Long) As Long

Public Declare Function CreateThread Lib "kernel32" _

  (ByVal lpThreadAttributes As Long, ByVal dwStackSize As Long, _

  ByVal lpStartAddress As Long, lpParameter As Long, _

  ByVal dwCreationFlags As Long, lpThreadId As Long) As Long

Public Declare Sub ExitThread Lib "kernel32" _�	(ByVal dwExitCode As Long)



Public Declare Function CreateEvent Lib "kernel32" _�	Alias "CreateEventA" _

  (ByVal lpEventAttributes As Long, ByVal bManualReset As Long, _

  ByVal bInitialState As Long, ByVal lpName As String) As Long

Public Declare Function ResetEvent Lib "kernel32" _�	(ByVal hEvent As Long) As Long

Public Declare Function SetEvent Lib "kernel32" _�	(ByVal hEvent As Long) As Long

Public Declare Function WaitForSingleObject Lib "kernel32" _

  (ByVal hHandle As Long, ByVal dwMilliseconds As Long) As Long

Public Declare Function WaitForMultipleObjects Lib "kernel32" _

  (ByVal nCount As Long, lpHandles As Long, _�	ByVal bWaitAll As Long, ByVal dwMilliseconds As Long) As Long

  

Public Const INFINITE = -1&

Public Const WAIT_TIMEOUT = &H102&

Public Const WAIT_OBJECT_0 = 0&

Public Const WAIT_FAILED = -1&



In einem BAS-Modul.

Weiter relevante API's , die hier nicht genutzt wurden : ResumeThread, SuspendThread, SetThreadPriority, TerminateThread, PulseEvent.

Wichtige Variablen-Deklarationen

' --- FishFace-Instanzen für die Robs

Public rb(1 To 4) As clsFishFace 

' --- Eventvariable zur Synchronisation der Robs 

Public rbEvent(1 To 4) As Long

' --- Eventvariable zur Fertigmeldung der Threads

Public rbFertig(1 To 4) As Long

' --- Thread-Handle für Rob-FahrAufträge

Public rbThread(1 To 4) As Long

' --- Thread ID 

Public rbThreadID(1 To 4) As Long

In einem BAS-Modul. Auch in Form eines Types möglich

�Initalisierung

Set rb(i) = New clsFishFace

' --- Erstellen der EventObjecte mit den Parametern :

'     Standard Security, Automatisches Reset des Events,

'     nicht signalisiert, keine Bennenung

rbEvent(i) = CreateEvent(ByVal 0&, 0&, 0&, vbNullString)

rbFertig(i) = CreateEvent(ByVal 0&, 0&, 0&, vbNullString)

Die EventObject bleiben, im Gegensatz zu den Threads (bei denen die CallBack-Routinen wechseln), über den Programmlauf erhalten

Standard Security : wird bei Win95/98 ignoriert.

Automatisches Reset : Zurücksetzen in den nicht signalisierten Status bei Abfrage durch WaitForSingleObject.

Auflösen

CloseHandle rbEvent(i)

CloseHandle rbFertig(i)

Set rb(i) = Nothing

Starten und Überwachen der Threads

' --- Zurücksetzen aller Events ---

ResetEvent rbEvent(i)

' --- Erstellen und Starten aller Rob-Threads ---

rbThread(i) = CreateThread(ByVal 0&, 0&, AddressOf DriveRob, _�    ByVal i, 0&, rbThreadID(i))

SetEvent rbEvent(3)           ' --- mit Rob3 starten ---

' --- Warten auf Ende aller Threads ---

Do

	DoEvents

Loop Until WaitForMultipleObjects(4&, rbFertig(1), True, 40&) _

               = WAIT_OBJect_0

' --- Schließen der Thread-Handles ---

CloseHandle rbThread(i)

Die CallBack-Routine (DriveRob) muß in einem BAS-Modul liegen. Übergabe der RobNr als Parameter (i), rbThreadID wird im Programm nicht genutzt.

Die Thread-Routine beginnt mit einem WaitForSingleObject

ResetEvent : eigentlich nicht erforderlich, Vorsichtsmaßnahme

SetEvent : Startfreigabe für Thread 3

WaitFor : Warten auf Ende aller (4) Threads (rbFertig(i) signaled), rbFertig(1) ist die Startadresse der Objectliste, Alle Threads müsen gleichzeitig signaled sein (True).�Gewartet wird jeweils 40 msec danach kehrt Wati mit WAIT_TIMEOUT zurück, erst wenn alle Thread signaled : WAIT_OBJECT_0. Diese Konstruktion wurde gewählt, um den Primär-Thread (die sichtbare Form) unterbrechbar zu halten (z.B. für Anzeigen aus den Threads). Waitxxx entspricht der Sleep-Funktion, sie parkt den Thread. Das entlastet die CPU, läßt aber keine Reaktionen des Threads mehr zu, die 40msec wurden recht willkürlich gewählt.

Nach Beendigung der Threads werden auch ihre Handles geschlossen, da sie ggf. gar nicht oder mit anderen CallBack-Routinen weiter genutzt werden.

�Thread-Routine

Public Function DriveRob(ByVal r As Long) As Long



' --- Betreiben eines Robs nach ProcTab unter Threadbedingungen ---

'     Ende : chkWieder oder Nothalt



Dim s&



	ResetEvent rbFertig(r)          ' --- Thread läuft

   

  Do

' --- Warten auf Start-Signal aus anderem Thread

    WaitForSingleObject rbEvent(r), INFINITE

	  If rbNotHalt Then Exit Do

' --- Abarbeiten des Script-Files mit den Fahraufträgen

    For s = ProcStart(r) To ProcStart(r + 1) - 1

		 If rbNotHalt Then Exit Do

      Select Case ProcTab(s).StepTyp

      Case 1                    ' --- Fahren ---

			 rb(r).MoveTo ProcTab(s).Wert1, ProcTab(s).Wert2, _

					ProcTab(s).Wert3, ProcTab(s).Wert4

		 Case 3                    ' --- Event ---

        If ProcTab(s).Wert1 = r Then

' --- Parken des eigenen Threads		

				 WaitForSingleObject rbEvent(r), INFINITE

        Else

          If ProcTab(s).Wert2 = 1 Then

' --- Startsignal für fremden Thread 

            SetEvent rbEvent(ProcTab(s).Wert1)

          Else

' --- Parksignal für fremden Thread

            ResetEvent rbEvent(ProcTab(s).Wert1)

          End If

        End If

      End Select

    Next s

    

' --- Endlos-Loop bis Abbruch durch Bediener

	Loop While frmRobs.chkWieder.Value = 1

  

  DriveRob = r

' --- Endesignal eigener Thread

  SetEvent rbFertig(r)

' --- Beenden eigener Thread

  ExitThread ByVal r

  

End Function

�Der Thread bezeichnet sich selber als busy (ResetEvent) und wartet dann auf seine Aktivierung von außen (WaitForSingleObject)

Der Thread läuft endlos, Beendigung von außen (Button Ende : SetEvent rbEvent und rbNothalt)

Der Thread arbeitet eine ScriptTabelle mit Arbeitsaufträgen ab : �rb(i).MoveTo : Fahrauftrag, Zugriff auf eine public Instanz der Klasse clsFishFace für den jeweiligen Robot. Achtung : konkurrierende Zugriffe (z.B. aus dem Primär-Thread) auf die Instanz sind nicht zulässig.�Set/ResetEvent : Synchronisation mit den anderen Robot-Threads. Die Plazierung dieser Synchronisationspunkte im Script-File muß sorgfältig erfolgen, da sich sonst die Robots gegenseitig stören, ggf. sind Hilfspositionen einzufügen nach deren Erreichen dann die Synchronisation erfolgt.

SetEvent rbFertig(r) : der Thread signalisiert sein eigenes Ende. Ein WaitMultiple direkt auf das ThreadEnde (ausgelöst durch ExitThread ist auch möglich, aber hier wegen der gegeseitigen Synchronisation problematisch).

Unter VB(5) sind innerhalb nicht alle VB-Befehle oder OCX-Methoden einsetzbar. Eine Dokumentation dazu liegt nicht vor. Möglich sind neben den hier gezeigten Befehlen auch Ausgaben in die Form des Primär-Threads (frmRob.Status = "LÄUFT"). Dazu muß muß der Primär-Thread natürlich leben (s.o.). Ebenso kann die eigentliche Thread-Routine im Code der Main-Form liegen (frmRob.subXXX) z.B. um auf die Form-Globalen Variablen und die Private-Routinen zugreifen zu können. Aber Achtung Statische Variable (das sind auch Variable auf Form-Ebene) werden im Gegensatz zu den dynamisch angelegten Variablen in den Subs gemeinsam genutzt.

